

Livret Utilisateur

Formules de calcul



SOMMAIRE

PRINCIPES DE BASE.....	3
NOTIONS D'ADRESSE ABSOLUE.....	3
NOMMER DES CELLULES.....	5
LES FONCTIONS MATHÉMATIQUES	6
LA FONCTION =SOMME()	6
LA FORMULE =SOMME.SI().....	6
LA FORMULE =SOMME.SI.ENS()	7
LA FORMULE =SOMMEPROD().....	8
LA FONCTION =MOYENNE()	10
LA FONCTION =MAX().....	10
LA FONCTION =MIN()	10
LA FONCTION =GRANDE.VALEUR()	11
LA FONCTION =PETITE.VALEUR().....	11
LA FONCTION =ARRONDI()	12
LA FONCTION =ARRONDI.INF().....	14
LA FONCTION =ARRONDI.SUP().....	15
LA FONCTION =ARRONDI.AU.MULTIPLE ().....	16
LES FONCTIONS DE DATE ET HEURE.....	17
LA FONCTION =AUJOURDHUI()	17
LA FONCTION =JOUR()	17
LA FONCTION =MOIS()	18
LA FONCTION =ANNEE().....	18
LA FONCTION =DATE()	19
LA FONCTION =JOURSEM()	19
LA FONCTION =DATEDIF()	20
LA FORMULE =NB.JOURS.OUVRES()	21
LA FORMULE =SERIE.JOUR.OUVRE()	22
LA FORMULE =FIN.MOIS().....	23
LA FORMULE = JOURS360().....	23
LES FONCTIONS DE TEXTE.....	24
LA FORMULE =MAJUSCULE().....	24
LA FORMULE =MINUSCULE().....	24
LA FORMULE =NOMPROPRE().....	25
LA FORMULE =CNUM().....	25
LES FONCTIONS LOGIQUES.....	26
LA FONCTION =SI().....	26
LA FONCTION ET()	27
LA FONCTION OU().....	28
LES FONCTIONS DE RECHERCHE.....	29
LA FORMULE =RECHERCHEV()	29
LA FONCTION =CHOISIR()	32
LES FONCTIONS STATISTIQUES.....	33
LA FONCTION =NB.SI()	33
LA FONCTION =NB.SI.ENS().....	34
1 ^{er} exemple :	34
2 ^{eme} exemple :.....	35
LA FORMULE =NBVAL()	36
LES FONCTIONS FINANCIÈRES.....	37
LA FONCTION =AMORLIN().....	37

ACTIVATION DES MACROS COMPLEMENTAIRES "UTILITAIRE D'ANALYSE".	38
POUR EXCEL XP – 2003	38
POUR EXCEL 2007	38

www.admexcel.com

Principes de base

Notions d'Adresse Absolue

Dans une formule de calcul il est fait référence, en général, à d'autres cellules.

Prenons le tableau suivant :

	A	B	C	D
1				
2				
3		Taux de remise	10%	
4				
5			Prix unitaire	Montant de la remise
6		Article 1	100	10
7		Article 2	300	0
8		Article 3	500	#VALEUR!
9				

Formule : C3*C6
en fait : L(-3)C(-1)

Après recopie vers le bas
on obtient : C4*C7

Après recopie vers le bas
on obtient : C5*C8

Si l'on veut dans ce tableau obtenir un prix après remise dans la cellule **D6**, on construit dans cette cellule la formule **=C6*C3**.

Excel raisonne sur les références de cellules utilisées. Pour lui, **C6** est la cellule qui se trouve sur la même ligne et dans la colonne de gauche par rapport à la cellule **D6** où l'on a construit la formule.

La référence utilisée est donc pour lui : **LC(-1)** qui veut dire : même ligne " L ", une colonne à gauche " C(-1)". De même, la référence **C3**, est pour lui **L(-3)C(-1)** qui veut dire trois lignes au-dessus " L(-3) ", une colonne à gauche " C(-1) "

Si l'on recopie cette formule sur les cellules D7 et D8, on va en fait recopier non pas la formule **=C6*C3**, mais la formule **=LC(-1)*L(-3)C(-1)**.

Le résultat obtenu n'est évidemment pas satisfaisant.

En effet, si l'on se place sur la cellule D8, on pourra lire dans la barre de formule le calcul suivant : **=C8*C5**.

C8 est la cellule contenant le prix unitaire voulu, mais **C5** ne contient pas le taux de remise mais un texte.

Ceci s'est produit car la formule recopiée en **D8** est celle construite en **D6**, c'est à dire : **=LC(-1)*L(-3)C(-1)**.

Par rapport à D8, **LC(-1)** se trouve être la cellule **C8**. Cette référence est correcte. Mais, par rapport à D8, **L(-3)C(-1)** se trouve être la cellule **C5**. Or, la cellule contenant le taux de

remise est la cellule **C3**. Notre calcul est donc faux. On dit alors qu'Excel travaille en fonction de références relatives.

Si l'on veut que la référence à la cellule C3 demeure lors d'une recopie de la formule de calcul, il faut neutraliser ce principe de référence relative. Pour cela, lors de la construction de la formule =C6*C3, il faut demander à ce la référence C3 soit absolue.

- Dans la barre de formule sélectionner C3
- Appuyer sur la touche F4

Constater le résultat dans la barre de formule. On obtient **=C6*\$C\$3**. La cellule C3 est devenue une référence absolue. Les symboles \$ (dollars) vont figer la référence.

Si l'on recopie la formule vers le bas en D7 et D8, on obtiendra :

- en D7 : C7*\$C\$3
- en D8 : C8*\$C\$3

Les symboles " \$ " peuvent être également ajoutés au clavier.

www.admexcel.com

Nommer des cellules


Dans une formule on utilise, en principe, le nom par défaut des cellules (A1, H25, B14, etc.).

Dans le but d'utiliser ces cellules dans un calcul, il est possible de les renommer.

En reprenant l'exemple précédent, on peut renommer la cellule C3 afin de l'utiliser ensuite dans les calculs.

Exemple : on va renommer la cellule C3 et lui donner le nom " Taux_remise ". Pour cela :

- on sélectionne la cellule C3 (son nom apparaît dans la " zone nom " à gauche de la barre de formule),
- on clique dans cette zone,
- on écrit le nom désiré (attention, pas d'espaces ni de caractères spéciaux tels que : / , . ; etc.)
- on valide avec la touche " Entrée "



A présent, toujours en reprenant l'exemple précédent, la formule de calcul pour la remise sera :

en D7 : **C7*Taux_remise**

Il est ensuite possible de recopier cette cellule vers le bas qui deviendra en D8 :

=C8*Taux_remise

Les fonctions mathématiques

La fonction =SOMME()

Pour additionner entre elles plusieurs cellules contiguës, on peut utiliser la fonction SOMME d'EXCEL.

- Ecrire dans la cellule où doit apparaître le résultat : =somme(
- Sélectionner ensuite à la souris la plage de cellules à additionner.
- Valider avec la touche Entrée

La fonction renvoie l'addition des sommes sélectionnées.

Exemple :

	A
1	12
2	-8
3	42
4	-15
5	

En A5 on écrit la formule suivante :
=SOMME(A1:A4)

Le résultat donne : **31**

La formule =SOMME.SI()

Cette fonction vous permet d'additionner plusieurs cellules en fonction d'un critère déterminé.

Voici sa syntaxe :

=SOMME.SI(plage;critère;somme_plage)

- plage est la plage de cellule contenant le critère à retenir
- critère est LE critère à retenir (chiffre ou texte)
- somme_plage est la plage de cellule où se trouvent les nombres à additionner.

Exemple : Dans la colonne A on a écrit des nombres. En colonne B, on a indiqué "garçon" ou "fille". On voudrait obtenir le total des nombres de la colonne A auxquels correspondent le mot "fille" en colonne B.

	A	B
1	26	Garçon
2	12	Fille
3	32	Fille
4	16	Garçon
5	22	Fille
6	48	Garçon
7		

En A7 écrivez la formule suivante :
=SOMME.SI(B1:B6;"fille";A1:A6)

Le résultat donne : **66**

La formule =SOMME.SI.ENS()

Cette formule permet d'additionner plusieurs cellules en fonction de plusieurs critères.

Voici sa syntaxe :

=SOMME.SI.ENS(Plage_somme ;**Plage_critères1**;Critères1;[**plage_critères2**;critères2];....)

Exemple :

	A	B	C	D	E	F	G
1	AUSTIN	Homme	COM	1 800			Tec
2	BOURD	Femme	TEC	1 300		homme	3330
3	CELAIRE	Femme	ADM	1 450			
4	CHINOIS	Homme	COM	1 620			
5	DELON	Femme	COM	1 930			
6	DEPARD	Femme	FIN	2 720			
7	DIAL	Homme	ADM	1 845			
8	NEMAR	Homme	ADM	1 730			
9	NOUGA	Homme	TEC	1 640			
10	POURUN	Homme	COM	1 550			
11	TERIEUR	Femme	COM	1 620			
12	TRAINE	Femme	TEC	1 480			
13	TUDEIL	Homme	COM	1 770			
14	ZIER	Homme	TEC	1 690			
15	ZOULOU	Homme	ADM	1 530			
16	ADJANI	Homme	ADM	1 710			
17	CHARME	Femme	FIN	1 980			
18	DAT	Femme	TEC	1 650			
19	DEJEAN	Homme	ADM	1 780			
20	DELUNE	Homme	COM	1 920			
21	DERAND	Homme	ADM	1 810			

=SOMME.SI.ENS(D:D;C:C;G1;B:B;F2)

Ici :

- Plage_somme, la plage de cellule contenant les montant à additionner, (dans la formule **D:D**),
- Première plage de critères ; la colonne du service, la colonne C (dans la formule **C:C**),
- Premier critère, le service saisi en cellule **G1**,
- Deuxième plage de critères, la colonne du sexe, la colonne B (dans la formule **B:B**),
- deuxième critère, le sexe saisi en cellule **F2**,

La formule =SOMMEPROD()

Réalisation d'opérations conditionnelles comptant et additionnant en utilisant des conditions multiples

Les formules permettant de calculer le nombre de valeurs ou les sommes sont fondées sur deux conditions ou plus.

La fonction que nous allons utiliser est SOMMEPROD ; celle-ci appartient à la catégorie Math & Trigo.

Exemple : Compter le nombre de mois où un commercial fait une vente. Le tableau de départ pourrait se présenter sous cette forme :

	A	B	C
1	JULES	Janvier	475
2	JULES	Janvier	110
3	MAXIME	Janvier	625
4	MAXIME	Janvier	235
5	PASCAL	Janvier	250
6	MAXIME	Janvier	550
7	PASCAL	Février	600
8	JULES	Février	750
9	MAXIME	Février	340
10	MAXIME	Février	190
11	MAXIME	Février	220
12	JULES	Février	400
13			

Exemple 1 : Compter le nombre des ventes faites par Jules sur le mois de février. C'est une formule conditionnelle qui emploie deux conditions.

= SOMMEPROD((A2:A13="Jules")*(B2:B13="Février"))

Le résultat donne : **2**

Exemple 2 : Compter les mois où les ventes sont plus grandes qu'une quantité spécifique

Vous pouvez employer une formule semblable à celle ci-dessus pour travailler avec des données numériques. Cette formule renvoie le nombre de lignes pour lesquelles le mois est égal à janvier et le montant est supérieur à 200.

= SOMMEPROD((B2:B13="Janvier")*(C2:C13>200))

Le résultat donne : **5**

Exemple 3 : Compter le nombre de ventes comprises entre certaines valeurs spécifiques

Cette formule compte le nombre de fois où les ventes sont comprises entre 200 et 500. La formule emploie deux conditions : la quantité supérieure ou égale à 200, et la quantité inférieure ou égale à 500.

=SOMMEPROD((C2:C13>=200)*(C2:C13<=500))

Le résultat donne : **6**

Exemple 4 : Compter le nombre de ventes basées sur trois conditions

Cette formule renvoie le nombre de fois où le commercial est Jules, le mois est février, et la quantité est supérieure à 500. Cette fonction emploie un argument simple, qui se compose de trois expressions logiques multipliées les unes aux autres.

=SOMMEPROD((A2:A13="Jules")*(B2:B13="Février")*(C2:C13>500))

Le résultat donne : **1**

Exemple 5 : Totaliser les ventes d'un commercial pour un mois donné

La formule calcule la somme des ventes basée sur deux conditions : quand le commercial est Jules et le mois est février.

=SOMMEPROD((A2:A13="Jules")*(B2:B13="Février")*(C2:C13))

Le résultat donne : **1150**

Exemple 6 : Totaliser les ventes combinées pour deux commerciaux

La formule ci-dessous renvoie la somme des ventes pour Jules et Pascal. Cette formule emploie un argument simple et se compose de trois expressions.

=SOMMEPROD(((A2:A13="Jules")+(A2:A13="Pascal"))*(C2:C13))

Le résultat donne : **2585**

Attention : les deux premiers arguments ne sont pas multipliés comme dans les exemples précédents. Dans la mesure où nous réalisons un test logique " OU ", ils sont additionnés. Addition de deux résultats logiques de valeur 1 quand l'une ou l'autre des expressions est VRAIE.

La fonction =MOYENNE()

Pour obtenir la moyenne de plusieurs cellules contiguës, on peut utiliser la fonction MOYENNE d'EXCEL.

- Ecrire dans la cellule où doit apparaître le résultat : =moyenne(
- Sélectionner ensuite à la souris la plage de cellules désirée
- Valider avec la touche Entrée.

La fonction renvoie la moyenne des sommes sélectionnées

Exemple : Dans la colonne A on a écrit des nombres. On voudrait obtenir la moyenne des nombres de la colonne A.

En A7 on écrit la formule suivante :
=MOYENNE(A1:A6)

Le résultat donne : **26**

	A
1	28
2	12
3	32
4	16
5	20
6	48

La fonction =MAX()

Pour obtenir le plus grand nombre contenu dans des cellules différentes, on peut utiliser la fonction MAX() d'EXCEL.

Voici sa syntaxe : =MAX(plage_de_cellules)

plage_de_cellules est la zone dans laquelle la valeur la plus élevée doit être cherchée.

- Ecrire dans la cellule où doit apparaître le résultat : =max(
- Sélectionner ensuite à la souris la plage de cellules (ou plusieurs plages de cellules en maintenant la touche Ctrl enfoncée) désirée.
- Valider avec la touche Entrée

la fonction renvoie la valeur la plus élevée

La fonction =MIN()

Pour obtenir le plus petit nombre contenu dans des cellules différentes, on peut utiliser la fonction MIN() d'EXCEL.

Voici sa syntaxe : =MIN(plage_de_cellules)

plage_de_cellules est la zone dans laquelle la valeur la moins élevée doit être cherchée.

- Ecrire dans la cellule où doit apparaître le résultat : =min(
- Sélectionner ensuite à la souris la plage de cellules (ou plusieurs plages de cellules en maintenant la touche Ctrl enfoncée) désirée.
- Valider avec la touche Entrée

La fonction =GRANDE.VALEUR()

Pour obtenir le plus petit nombre contenu dans des cellules différentes, on peut utiliser la fonction GRANDE.VALEUR() d'EXCEL.

Voici sa syntaxe :

=GRANDE.VALEUR(*plage_de_cellules*;rang à prendre en compte)

plage_de_cellules est la zone dans laquelle la valeur doit être cherchée.

rang à prendre en compte est le nombre correspondant au rang de la valeur cherchée.

Si vous écrivez 3 la fonction cherchera la 3eme plus grande valeur de la *plage_de_cellules*.

- Ecrire dans la cellule où doit apparaître le résultat : =grande.valeur(
- Sélectionner ensuite à la souris la *plage de cellules* (ou plusieurs plages de cellules en maintenant la touche Ctrl enfoncée) désirée.
- écrivez ; (point virgule) puis le **rang** souhaité
- Valider avec la touche Entrée

la fonction renvoie la valeur souhaitée

La fonction =PETITE.VALEUR()

Pour obtenir le plus petit nombre contenu dans des cellules différentes, on peut utiliser la fonction PETITE.VALEUR() d'EXCEL.

Voici sa syntaxe :

=PETITE.VALEUR(*plage_de_cellules*;rang à prendre en compte)

plage_de_cellules est la zone dans laquelle la valeur doit être cherchée.

rang à prendre en compte est le nombre correspondant au rang de la valeur cherchée.

Si vous écrivez 3 la fonction cherchera la 3eme plus petite valeur de la *plage_de_cellules*.

- Ecrire dans la cellule où doit apparaître le résultat : =petite.valeur(
- Sélectionner ensuite à la souris la *plage de cellules* (ou plusieurs plages de cellules en maintenant la touche Ctrl enfoncée) désirée.
- écrivez ; (point virgule) puis le **rang** souhaité
- Valider avec la touche Entrée

la fonction renvoie la valeur souhaitée

La fonction =ARRONDI()

Le résultat d'une division ou d'une multiplication peut donner un nombre comportant plusieurs décimales. Dans le but de créer des formulaires destinés à des tiers (exemple facture) les nombres doivent, en principe, se présenter avec simplement deux décimales. On peut, bien entendu, utiliser le format des nombres. Mais Excel va conserver en mémoire le nombre initial avec plusieurs décimales et l'utilisera pour les calculs ultérieurs. D'où des problèmes d'affichages.

Exemple :

Le résultat du calcul : 2 divisé par 3 donne le résultat 0.66666666.....

Si on additionne 0.66666666.... avec 0.66666666.... on obtient 1.33333333.....

Avec un format des nombres à deux décimales, Excel propose un arrondi de présentation qui donnera : $0.67 + 0.67 = 1.33$. Evidemment, cet arrondi de présentation n'est pas satisfaisant. Il faut donc utiliser une formule d'EXCEL, la formule ARRONDI()

Voici sa syntaxe :

=ARRONDI(*nombre*;No_chiffres)

Cette formule requiert deux arguments :

nombre est le nombre à arrondir. Ce nombre peut également être remplacé par la référence à une cellule contenant un nombre. Il peut également être une formule de calcul complexe.

No_chiffres représente le nombre de décimales que l'on souhaite obtenir.

Exemple : pour obtenir le résultat du calcul 2/3 arrondi à 2 décimales :

=ARRONDI(2/3;2)

Le résultat de cette formule sera 0.67. Ce nombre s'affichera directement dans la cellule sans que l'on ait besoin de passer par le format des nombres.

L'argument "nombre de décimales " donne des arrondis différents selon le chiffre utilisé.

Voici un exemple des arrondis possibles du nombre 10734,12 :

Pour obtenir un nombre arrondi, on peut utiliser la fonction ARRONDI() d'EXCEL.

Voici sa syntaxe :

=ARRONDI(*nombre*;no_chiffres)

nombre représente le nombre à arrondir.

no_chiffres spécifie le nombre de chiffres auquel vous voulez arrondir nombre.

- Si **no_chiffres** est supérieur à 0 (zéro), nombre est arrondi au nombre de décimales indiqué.
- Si **no_chiffres** est égal à 0, nombre est arrondi au nombre entier le plus proche.
- Si **no_chiffres** est inférieur à 0, nombre est arrondi à gauche de la virgule.

Exemple :

	A	B	C
1	10734.12	=ARRONDI(A1;2)	10734.12
2	10734.12	=ARRONDI(A2;1)	10734.1
3	10734.12	=ARRONDI(A3;0)	10734
4	10734.12	=ARRONDI(A4;-1)	10730
5	10734.12	=ARRONDI(A5;-2)	10700
6	10734.12	=ARRONDI(A6;-3)	11000

La fonction =ARRONDI.INF()

Pour obtenir un nombre arrondi tendant vers 0 (zéro).

Voici sa syntaxe :

=ARRONDI.INF(*nombre*;no_chiffres)

nombre représente un nombre réel quelconque à arrondir en tendant vers zéro.

no_chiffres représente le nombre de chiffres à prendre en compte pour arrondir l'argument nombre.

La fonction =ARRONDI.INF est similaire à la fonction ARRONDI, excepté qu'elle arrondit toujours le nombre en tendant vers zéro.

Si l'argument **no_chiffres** est supérieur à 0 (zéro), le nombre est arrondi à la valeur entière immédiatement inférieure (ou supérieure pour les nombres négatifs) et comporte le nombre de décimales spécifié.

Si l'argument **no_chiffres** est égal à 0, le nombre est arrondi au nombre entier immédiatement inférieur.

Si l'argument **no_chiffres** est inférieur à 0, le nombre est arrondi à la valeur immédiatement inférieure (ou supérieure si négative) par incréments de 10, 100, etc., en fonction de la valeur de **no_chiffres**.

Exemple :

Formule	Description (résultat)
=ARRONDI.INF(3,2;0)	Arrondit 3,2 à la valeur entière immédiatement inférieure (3)
=ARRONDI.INF(76,9;0)	Arrondit 76,9 à la valeur entière immédiatement inférieure (76)
=ARRONDI.INF(3,14159;3)	Arrondit 3,14159 à la valeur inférieure comportant trois décimales (3,141)
=ARRONDI.INF(-3,14159;1)	Arrondit -3,14159 à la valeur inférieure comportant une décimale (-3,1)
=ARRONDI.INF(31415,92654;-2)	Arrondit 31415,92654 à gauche du séparateur décimal, à la centaine immédiatement inférieure (31400)

La fonction =ARRONDI.SUP()

Pour obtenir un nombre arrondi s'éloignant de 0 (zéro).

Voici sa syntaxe :

=ARRONDI.SUP(*nombre*;no_chiffres)

nombre représente un nombre réel quelconque à arrondir en s'éloignant de zéro.

no_chiffres représente le nombre de chiffres à prendre en compte pour arrondir l'argument nombre.

La fonction ARRONDI.SUP est similaire à la fonction ARRONDI, excepté qu'elle arrondit toujours le nombre en s'éloignant de zéro.

Si l'argument **no_chiffres** est supérieur à 0 (zéro), le nombre est arrondi à la valeur immédiatement supérieure (ou inférieure pour les nombres négatifs) et comporte le nombre de décimales spécifié.

Si l'argument **no_chiffres** est égal à 0 ou omis, le nombre est arrondi au nombre entier immédiatement supérieur.

Si l'argument **no_chiffres** est inférieur à 0, le nombre est arrondi à la valeur immédiatement supérieure (ou inférieure si négative) par incréments de 10, 100, etc., en fonction de la valeur de **no_chiffres**.

Exemple

Formule	Description (résultat)
=ARRONDI.SUP(3,2;0)	Arrondit 3,2 à la valeur entière immédiatement supérieure (4)
=ARRONDI.SUP(76,9;0)	Arrondit 76,9 à la valeur entière immédiatement supérieure (77)
=ARRONDI.SUP(3,14159;3)	Arrondit 3,14159 à la valeur supérieure comportant trois décimales (3,142)
=ARRONDI.SUP(-3,14159;1)	Arrondit -3,14159 à la valeur supérieure comportant une décimale (-3,2)
=ARRONDI.SUP(31415,92654;-2)	Arrondit 31415,92654 à gauche du séparateur décimal, à la centaine immédiatement supérieure (31500)

La fonction =ARRONDI.AU.MULTIPLE ()

Pour obtenir un nombre au multiple spécifié.

Si cette fonction n'est pas disponible et renvoie la valeur d'erreur #NOM?, installez et chargez la macro complémentaire Utilitaire d'analyse.

Voici sa syntaxe :

=ARRONDI.AU.MULTIPLE(*nombre*;**multiple**)

nombre représente la valeur à arrondir.

multiple représente le multiple auquel vous souhaitez arrondir nombre.

ARRONDI.AU.MULTIPLE arrondit en s'éloignant de zéro, si le reste de la division de nombre par multiple est supérieur ou égal à la moitié de la valeur de multiple.

Exemple

Formule	Description (résultat)
=ARRONDI.AU.MULTIPLE(10;3)	Arrondit 10 au multiple de 3 le plus proche (9)
=ARRONDI.AU.MULTIPLE(-10;-3)	Arrondit -10 au multiple de -3 le plus proche (-9)
=ARRONDI.AU.MULTIPLE(1,3;0,2)	Arrondit 1,3 au multiple de 0,2 le plus proche (1,4)
=ARRONDI.AU.MULTIPLE(5;-2)	Renvoie une erreur, car 5 et -2 sont de signes opposés (#NOMBRE!)

Les fonctions de date et heure

La fonction =AUJOURDHUI()

Cette fonction vous permet d'afficher automatiquement la date du système (ordinateur) dans une cellule.

Vous pouvez également utiliser cette fonction en combinaison avec une autre formule de calcul sur les dates.

Voici sa syntaxe :

=AUJOURDHUI()

Attention : Il n'y a rien à écrire entre les parenthèses

Exemple : Vous avez saisi une date d'échéance dans la cellule A1 (par exemple 12/06/04). En B1 vous désirez connaître le nombre de jours restant entre cette date d'échéance et la date du jour. En B1 vous écrivez :

=A1-AUJOURDHUI()

En supposant que la date du jour soit le 3 juin 2004, Le résultat est : **9**

La fonction =JOUR()

Cette fonction prend tout son intérêt dans le cadre de la fonction =DATE() décrite ci-après. Mais elle peut aussi être utilisée seule.

Cette fonction vous permet d'extraire le numéro du jour du mois à partir d'une date saisie dans une cellule (ou dans la formule elle même). Le numéro extrait peut ainsi être utilisé pour effectuer des calculs ou des test dans le cadre d'une fonction logique.

Vous pouvez également utiliser cette fonction en combinaison avec une autre formule de calcul sur les dates.

Voici sa syntaxe :

=JOUR()

Exemple : Vous avez saisi une date de paiement dans la cellule A1 (par exemple 12/06/04). En B1 vous désirez connaître le jour où la facture devra être payée compte tenu d'un délai de paiement de 8 jours. En B1 vous écrivez :

=JOUR(A1)+8

Le résultat est : **23/06/04** (ou 23 si vous modifiez le format des nombres)

La fonction =MOIS()

Cette fonction prend tout son intérêt dans le cadre de la fonction =DATE() décrite ci-après. Mais elle peut aussi être utilisée seule.

Cette fonction vous permet d'extraire le numéro du mois à partir d'une date saisie dans une cellule (ou dans la formule elle-même). Le numéro extrait peut ainsi être utilisé pour effectuer des calculs ou des tests dans le cadre d'une fonction logique.

Vous pouvez également utiliser cette fonction en combinaison avec une autre formule de calcul sur les dates.

Voici sa syntaxe :

=MOIS()

Exemple : Vous avez saisi une date de paiement dans la cellule A1 (par exemple 12/06/04). En B1 vous désirez connaître le jour où la facture devra être payée compte tenu d'un délai de paiement de 2 mois. En B1 vous écrivez :

=MOIS(A1)+1

Le résultat est : **12/07/04** (ou 07 si vous modifiez le format des nombres)

La fonction =ANNEE()

Cette fonction prend tout son intérêt dans le cadre de la fonction =DATE() décrite ci-après. Mais elle peut aussi être utilisée seule.

Cette fonction vous permet d'extraire le numéro de l'année à partir d'une date saisie dans une cellule (ou dans la formule elle-même). Le numéro extrait peut ainsi être utilisé pour effectuer des calculs ou des tests dans le cadre d'une fonction logique.

Vous pouvez également utiliser cette fonction en combinaison avec une autre formule de calcul sur les dates.

Voici sa syntaxe :

=ANNEE()

Exemple : Vous avez saisi une date la cellule A1 (par exemple 12/06/04). En B1 vous désirez ajouter une année à cette date. En B1 vous écrivez :

=ANNEE(A1)+1

Le résultat est : **12/06/05** (ou 2005 si vous modifiez le format des nombres)

La fonction =DATE()

Cette fonction peut se combiner avec les fonctions ci-dessus. Mais elle peut aussi être utilisée seule.

Cette fonction vous permet de calculer une date à partir d'une autre

Vous pouvez également utiliser cette fonction en combinaison avec une autre formule de calcul sur les dates.

Voici sa syntaxe :

=DATE(annee;mois;jour)

Exemple : Vous avez saisi une date de paiement dans la cellule A1 (par exemple 12/06/04). En B1 vous désirez connaître le jour où la facture devra être payée compte tenu d'un délai de paiement à 30 jours. En B1 vous écrivez :

=DATE(ANNEE(A1);MOIS(A1)+1;JOUR(A1))

Le résultat est : **12/07/04**

Exemple 2 : Vous voulez calculer un amortissement et vous avez besoin d'obtenir automatiquement la date du dernier jour de l'exercice comptable. Vous avez saisi la date d'achat de matériel dans la cellule A1 (par exemple 15/06/04). En B1 vous désirez obtenir la date de fin d'exercice (à savoir le 31 décembre de l'année). En B1 vous écrivez :

=DATE(ANNEE(A1);12;31)

Le résultat est : **31/12/04**

La fonction =JOURSEM()

Cette fonction vous permet de connaître le numéro du jour de la semaine à partir d'une date saisie dans une cellule (ou dans la formule elle-même). Le numéro extrait peut ainsi être utilisé pour effectuer des calculs ou des tests dans le cadre d'une fonction logique. Vous pouvez également utiliser cette fonction en combinaison avec une autre formule de calcul sur les dates.

Voici sa syntaxe :

=JOURSEM(Numéro_de_serie;type_de_retour)

Numéro_de_serie est la date de référence (saisie ou référence à une cellule)

type_de_retour est la codification des jours :

- code 1 : dimanche = 1 et samedi = 7
- code 2 : lundi = 1 et dimanche = 7
- code 3 : lundi = 0 et dimanche = 1

Exemple : Vous avez saisi une date de paiement dans la cellule A1 (par exemple 9/06/04). En B1 vous désirez connaître le jour de la semaine correspondant. En B1 vous écrivez :

=JOURSEM(A1;2)

Le résultat est : 3 (ou mercredi si vous modifiez le format des nombres)

La fonction =DATEDIF()

Attention, pour faire fonctionner cette formule vous devez activer la macro complémentaire "Utilitaire d'analyse". Voir page 38

La fonction =DATEDIF() permet de calculer la différence entre deux dates en années, mois et jours. Elle comporte 3 arguments :

Voici sa syntaxe :

=DATEDIF(Date1;Date2;code)

Cette fonction renvoie la différence entre Date1 et Date2 (attention, Date2 >= Date1) selon l'argument Code, qui peut prendre les valeurs suivantes :

- "y" : différence en années
- "m" : différence en mois
- "d" : différence en jours
- "ym" : différence en mois, une fois les années soustraites
- "yd" : différence en jours, une fois les années soustraites
- "md" : différence en jours, une fois les années et les mois soustraits

La fonction =DATEDIF() peut être en particulier utilisée pour calculer des âges. Par exemple, si la cellule A1 contient une date de naissance et la cellule A2 la date du jour :

Exemple : en A1 : 20/04/1963 et en A2 : 27/05/2002

	Formule	Résultat
En années	=DATEDIF(A1;A2;"y")	39
En mois	=DATEDIF(A1;A2;"m")	469
En jours	=DATEDIF(A1;A2;"d")	14282
différence en mois, une fois les années soustraites	=DATEDIF(A1;A2;"ym")	1
différence en jours, une fois les années soustraites	=DATEDIF(A1;A2;"yd")	37
différence en jours, une fois les années et les mois soustraits	=DATEDIF(A1;A2;"md")	7

La formule =NB.JOURS.OUVRES()

Attention, pour faire fonctionner cette formule vous devez activer la macro complémentaire "Utilitaire d'analyse". Voir page 38

Cette formule vous permet d'obtenir le nombre de jours ouvrés entre 2 dates (Rappel : Il y a 5 jours ouvrés par semaine).

Voici sa syntaxe :

=NB.JOURS.OUVRES(date_départ;date_fin;jours_fériés)

Les dates de départ et de fin peuvent être saisies dans la formule ou faire référence à des cellules où elles ont été saisies.

Les jours fériés (optionnel) sont entrés dans la formule en faisant référence à une plage de cellules où ils ont été saisis.

Exemple :

On cherche le nombre de jours ouvrables entre le 1er janvier et le 30 juin 2010 :

- En A1 on entre la date de départ : 01/01/2010,
- En A2 on entre la date de fin : 30/06/2010,
- De A5 à A14 on entre les dates des jours fériés de 2010
- En C1 on écrit la formule suivante :

=NB.JOURS.OUVRES(A1;A2;A4:A14)

Le résultat donne : **124**

	A	B	C
1	01/01/2010		124
2	30/06/2010		
3			
4	01/01/2010		
5	05/04/2010		
6	01/05/2010		
7	08/05/2010		
8	13/05/2010		
9	24/05/2010		
10	14/04/2010		
11	15/08/2010		
12	01/11/2010		
13	11/11/2010		
14	25/12/2010		

=NB.JOURS.OUVRES(A1;A2;A4:A14)

La formule =SERIE.JOUR.OUVRE()

Attention, pour faire fonctionner cette formule vous devez activer la macro complémentaire "Utilitaire d'analyse". Voir page 38

Cette formule vous permet d'obtenir une date de fin en tenant compte de la date de départ, du nombre de jours ouvrés qui devront s'écouler, et des jours fériés compris entre ces 2 dates. (Rappel : Il y a 5 jours ouvrés par semaine).

Voici sa syntaxe :

=SERIE.JOUR.OUVRE(date_départ;nb_jours;jours_fériés)

Les dates de départ et le nombre de jours à ajouter peuvent être saisies dans la formule ou faire référence à des cellules où elles ont été saisies.

Les jours fériés (optionnel) sont entrés dans la formule en faisant référence à une plage de cellules où ils ont été saisis.

Exemple : Une facture établie le 10 mars 2010 devra être payée au bout de 60 jours ouvrables.

- En A1 on entre la date de départ : 10/03/2010,
- En A2 on entre la nombre de jours à ajouter : 60,
- De A5 à A14 on entre les dates des jours fériés de 2010
- En C1 on écrit la formule suivante :

=SERIE.JOUR.OUVRE(A1;A2;A4:A14)

Le résultat donne : **06/06/2010**

	A	B	C
1	10/03/2010		08/06/2010
2	60		
3			
4	01/01/2010		
5	05/04/2010		
6	01/05/2010		
7	08/05/2010		
8	13/05/2010		
9	24/05/2010		
10	14/04/2010		
11	15/08/2010		
12	01/11/2010		
13	11/11/2010		
14	25/12/2010		

=SERIE.JOUR.OUVRE(A1;A2;A4:A14)

La formule =FIN.MOIS()

Attention, pour faire fonctionner cette formule vous devez activer la macro complémentaire "Utilitaire d'analyse". Voir page 38

Cette formule vous permet d'obtenir le dernier jour d'un mois à partir d'une date. Utile pour calculer une échéance à 30 jours fin de mois par exemple.

Voici sa syntaxe :

=FIN.MOIS(date_départ;Mois)

Les dates de départ et le mois peuvent être saisies dans la formule ou faire référence à des cellules où elles ont été saisies.

Exemple : Une facture établie le 10 mars 2010 devra être payée avec une échéance 30 jours fin de mois.

- En A1 on entre la date de départ : 10/03/2010,
- En A2 on écrit la formule suivante :

=FIN.MOIS(A1;1)

Le résultat donne : **30/04/2010**

La formule = JOURS360()

Cette formule calcule le nombre de jours écoulés entre 2 dates sur la base d'une année de 360 jours (12 mois de 30 jours)

Voici sa syntaxe :

=JOURS360(date_début ;date_fin ;méthode)

date_début et **date_fin** sont les 2 dates entre lesquelles vous désirez compter le nombre de jours écoulés.

Méthode est la méthode de comptage retenue :

- **0** (zéro) ou omis, il s'agit de la méthode de comptage américaine (US (NASD)). Si la date de début est le 31 du mois, la date de début devient le 30 du même mois. Si la date de fin est le 31 du mois et que la date de début est avant le 30 du mois, la date de fin devient le 1er du mois suivant ; sinon, la date de fin devient le 30 du même mois.
- **1**, il s'agit de la méthode européenne. Les dates de début ou de fin correspondant au 31 du mois deviennent le 30 du même mois.

Exemple : En A1 on écrit : 02/01/10, et en A2 on écrit : 31/01/10.

- Si en A3 on écrit la formule : =JOURS360(A1 ;A2 ;0) le résultat donne : **29**
- Si en A3 on écrit la formule : =JOURS360(A1 ;A2 ;1) le résultat donne : **28**

Les fonctions de texte

La formule =MAJUSCULE()

Cette formule vous permet de convertir le texte d'une cellule en MAJUSCULE.

Voici sa syntaxe :

=MAJUSCULE(*Texte*)

Texte fait référence à UNE cellule où un texte apparaît pour tout ou partie en minuscules, et le convertit en majuscules.

Exemple : en A1 vous avez le texte "bonjour".

En B1 on écrit la formule suivante :

=MAJUSCULE(A1)

Le résultat donne : **BONJOUR**

Astuce :

Une fois la conversion réussie, vous pouvez copier/ coller le résultat obtenu dans la cellule d'origine (ici A1) en utilisant le "collage spécial/ options "valeurs".

La formule =MINUSCULE()

Cette formule vous permet de convertir le texte d'une cellule en MINUSCULE.

Voici sa syntaxe :

=MINUSCULE(*Texte*)

Texte fait référence à UNE cellule où un texte apparaît pour tout ou partie en majuscule, et le convertit en minuscule.

Exemple : en A1 vous avez le texte "SALUT".

En B1 on écrit la formule suivante :

=MINUSCULE(A1)

Le résultat donne : **salut**

Astuce :

Une fois la conversion réussie, vous pouvez copier/ coller le résultat obtenu dans la cellule d'origine (ici A1) en utilisant le "collage spécial/ options "valeurs".

La formule =NOMPROPRE()

Cette formule vous permet de mettre en MAJUSCULE la première lettre de CHAQUE mot d'un texte contenu dans une cellule.

Voici sa syntaxe :

=NOMPROPRE(*Texte*)

Texte fait référence à UNE cellule où un texte apparaît pour tout ou partie en minuscules, et le convertit en minuscules en ajoutant une majuscule au début de chaque mot.

Exemple : en A1 vous avez le texte "le matin suivant".

En B1 on écrit la formule suivante :

=NOMPROPRE(A1)

Le résultat donne : **Le Matin Suivant**

Astuce :

Une fois la conversion réussie, vous pouvez copier/ coller le résultat obtenu dans la cellule d'origine (ici A1) en utilisant le "collage spécial/ options "valeurs".

La formule =CNUM()

Cette formule vous permet de redonner un format de NOMBRE à une liste de chiffres importés à partir d'une base de donnée de type ACCES. En effet, il arrive que les séries de nombres importés soient au format TEXTE. Il est alors impossible de les utiliser dans Excel pour effectuer des calculs. Cette fonction vous permet de leur rendre leur format de nombre. Attention cela ne fonctionne pas avec des cellules Excel que vous auriez vous-même mis au format "Texte".

Voici sa syntaxe :

=CNUM(*Texte*)

Texte fait référence à une cellules où un nombre apparaît une chaîne textuelle représentant un nombre.

Exemple : en A1 vous avez le nombre 128. En demandant un format d'alignement standard, vous constatez qu'il est aligné sur le bord gauche de la cellule, signe qu'il s'agit d'un texte.

En B1 on écrit la formule suivante :

=CNUM(A1)

Le résultat donne : **128** au format de nombre utilisable

Astuce :

Une fois la conversion réussie, vous pouvez copier/ coller le résultat obtenu dans la cellule d'origine (ici A1) en utilisant le "collage spécial/ options "valeurs".

Les fonctions logiques

La fonction =SI()

On peut demander à Excel de réagir différemment en fonction du résultat d'un calcul demandé.

Exemple :

Vous décidez d'élaborer une facture. Vous souhaitez faire bénéficier à votre client d'une remise sur le prix de vente. Cette remise sera de 5% du total hors taxes de la facture si ce total inférieur à 1000 €. Si le total hors taxes est supérieur ou égal à 1000 €, le taux de la remise sera de 10% sur la totalité du montant.

Le document Excel se présentera ainsi :

	A	B	C	D	E	F	G
1	Désignation	Prix unitaire	Quantité	Montant total			
2	Article 1	150,00				B2*C2	
3	Article 2	200,00				B3*C3	
4	Article 3	320,00				B4*C4	
5		Total hors taxes				=SOMME(D2:D4)	
6		Remise					
7							
8							
9							
10							
11							
12							
13							
14							
15							

=SI(D5<1000;D5*5%;D5*10%)

ALORS SINON

Comparaison (test) à effectuer

Action à faire si le résultat du test est vrai

Action à faire si le résultat du test est faux

La formule en D6, compare tout d'abord le résultat obtenu en D5 au nombre 1000.

Si le montant en D5 est plus petit que 1000, **alors** (symbolisé par le point virgule) le résultat à afficher en D6 sera celui du calcul : D5 multiplié par 5%.

Sinon (symbolisé par le deuxième point virgule) c'est à dire si le montant en D5 est égal ou plus grand que le nombre 1000, le résultat à afficher en D6 sera celui du calcul : D5 multiplié par 10 %

La fonction ET()

Les conditions posées dans le test de la fonction SI() seront multiples et pour que la réponse aux conditions soit "VRAI", il faudra qu'elles se vérifient toutes.

Voici sa syntaxe :

=SI(ET(Cond1;Cond2;...;CondN);action à réaliser si les N conditions sont satisfaites;action à réaliser si au moins une des conditions n'est pas satisfaite)

Exemple : On désire attribuer une ristourne de fin d'année de 2% aux clients qui remplissent les 2 conditions suivantes :

être grossiste

ET

avoir acheté plus de 100 000 € de marchandises chez nous.

Nos clients sont répertoriés dans le tableau ci-dessous.

	A	B	C
1	Nom du client	Type de client	Achats
2	EMELINE	Grossiste	67 000 €
3	ACQUIN	Détaillant	138 000 €
4	HENDOL	Grossiste	213 000 €
5	JUNEZ	Détaillant	59 500 €

On crée une colonne ristourne et on saisi dans cette colonne la formule suivante :

=SI(ET(type de client="grossiste";achats>100000);achats*2%;0)

Remarque : Cette formule peut être recopiée sur les lignes suivantes. On remarque que l'on utilise les nom des en-têtes de colonnes. Cette technique fonctionne indépendamment des fonctions SI(), ET() et OU().

	A	B	C	D
1	Nom du client	Type de client	Achats	Ristourne
2	EMELINE	Grossiste	67 000 €	0 €
3	ACQUIN	Détaillant	138 000 €	0 €
4	HENDOL	Grossiste	213 000 €	4 260 €
5	JUNEZ	Détaillant	59 500 €	0 €

La fonction OU()

Les conditions posées seront multiples et pour que la réponse aux conditions soit "VRAI", il faudra que l'une au moins se vérifie.

Voici sa syntaxe :

=SI(OU(Cond1;Cond2;...;CondN);action à réaliser si au moins une des conditions est satisfaite;action à réaliser si aucune des conditions n'est satisfaite)

Exemple : Une entreprise souhaite verser une prime de fin d'année à ses représentants s'ils remplissent l'une **OU** l'autre des conditions suivantes :

Avoir plus de 5 ans d'ancienneté

OU

Avoir réalisé plus d' 1 Million d'euros de chiffre d'affaires.

Les représentants sont répertoriés dans le tableau suivant :

	A	B	C
1	Nom du Représentant	Années d'ancienneté	Chiffre d'affaires
2	ARMAND	3	1 213 000
3	FLORA	6	997 000
4	NINIAN	7	1 016 000
5	SEBATI	1	757 000

On crée une colonne PRIME et on y construit la formule suivante :

=SI(OU(années d'ancienneté>5;chiffre d'affaires>1000000);1000;0)

Remarque : Cette formule peut être recopiée sur les lignes suivantes. On remarque que l'on utilise les nom des en-têtes de colonnes. Cette technique fonctionne indépendamment des fonctions SI(), ET() et OU().

	A	B	C	D
1	Nom du Représentant	Années d'ancienneté	Chiffre d'affaires	Prime
2	ARMAND	3	1 213 000	0 €
3	FLORA	6	997 000	0 €
4	NINIAN	7	1 016 000	4 260 €
5	SEBATI	1	757 000	0 €

Les fonctions de recherche

La formule =RECHERCHEV()

Cette fonction permet de renvoyer une information à partir d'un critère recherché dans un tableau, par exemple le prix d'un article à partir de sa référence.

Voici sa syntaxe :

=RECHERCHEV(**valeur_cherchée**;table_matrice;no_index_col;**valeur_proche**)

La fonction va rechercher, dans une *table_matrice*, la **valeur_cherchée** que l'on va lui indiquer. Quand elle aura trouvé cette valeur (nombre ou texte), elle va renvoyer le contenu de la cellule se trouvant sur la même ligne que la valeur trouvée et dans la colonne désignée par no_index_col.

Si la **valeur_cherchée** n'existe pas dans la *table_matrice*, la fonction affichera :

- si on choisi le nombre 0 pour l'argument **valeur_proche**, un message d'erreur,
- si on choisi le nombre 1 pour l'argument **valeur_proche**, le contenu de la colonne, déterminée par no_index_col, correspondant à la valeur précédente la plus proche de la **valeur_cherchée**.

Exemple :

Pour élaborer une facture, nous allons utiliser la fonction =RECHERCHEV().

Tout d'abord, nous construisons 3 tableaux, un sur chaque feuille du classeur, un modèle de facture, un tableau des articles (nous nommerons la zone « articles » pour la réutiliser dans les formules) et un tableau de remises (que nous nommerons « remise »).

	A	B	C	D	E
1	Code Article	Désignation	Quantité	Prix Unitaire	Montant Total
2					
3					
4					
5					
6					
7					
8					
9			TOTAL marchandises		
10				Remise	
11				TOTAL NET	
12				TVA	
13				TOTAL TTC	
14					

	A	B	C
1	Code article	Désignation	Prix unitaire
2	1	Disque dur	2000
3	2	Lecteur de disquette	300
4	3	souris	100
5	4	clavier	250
6	5	Alimentation	500
7	6	boitier	500
8	7	écran	1700

	A	B
1	total ht	taux
2	client de passage	0%
3	Client régulier	5%
4	client important	7%
5	Grossiste	10%
6	Très gros clients	12%
7	remise exceptionnelle	15%

La première formule à créer permettra d'afficher automatiquement la désignation de l'article (dans la colonne « désignation ») dont le code aura été saisi en A2 (colonne « code article»). Nous allons construire cette formule dans la colonne " Désignation " (en B2)

La formule, que l'on va construire dans la cellule B2, sera donc la suivante :

- **La valeur_cherchée** sera le code de l'article, préalablement saisi en A2.
- La *table_matrice* est la plage de cellules que l'on a préalablement nommé " articles ".
- *no_index_col* sera ici le chiffre 2 qui correspond à la deuxième colonne de notre zone " articles ", et qui contient le nom des articles.
- **valeur_proche** sera ici le chiffre 0 qui permettra d'afficher un message d'erreur si le code article saisi n'existe pas.

=RECHERCHEV(A2;articles;2;0)

Le résultat donne : **Disque dur.**

	A	B	C	D	E
1	Code Article	Désignation	Quantité	Prix Unitaire	Montant Total
2	1	Disque dur			
3					
4					
5					
6					
7					
8					
9		TOTAL marchandises			
10				Remise	
11				TOTAL NET	
12				TVA	
13				TOTAL TTC	
14					

En effet :

la fonction recherche la valeur 1, saisie en A2, dans la zone " articles " en se déplaçant VERTICALEMENT dans la première colonne de la zone.

une fois la **valeur_cherchée** trouvée, la fonction lit la ligne correspondante dans la zone jusqu'à la colonne choisie dans *no_index_col*.

la fonction renvoie la valeur de cette cellule.

Pour obtenir le prix de l'article, la fonction est la même que la précédente, mis à part le *no_index_col* qui est ici le chiffre 3 (troisième colonne de la zone " articles "). La fonction construite en D2 est la suivante :

=RECHERCHEV(A2;articles;3;0)

Pour obtenir le montant total, on multiplie la quantité en C2 par le prix obtenu en D2. la formule est la suivante :

=C2*D2

Toutes ces formules de la ligne 2 sont à recopier sur les lignes suivantes.

Le total hors taxes de la facture s'obtient en faisant la somme des lignes de la facture :

=SOMME(E2:E8)

Le montant de la remise sera fonction du total hors taxes de la facture situé en E9. On applique ici un barème par tranches. Par exemple, entre 5000 € et 7000 €, on applique 5% de remise. Dès que le montant hors taxes dépasse 7000 € et jusqu'à ce qu'il atteigne 10000 €, le taux de remise sera de 7%, etc.

Dans notre tableau des remises, tous les cas de figure ne sont pas prévus. Seuls les seuils sont représentés. la fonction de recherche se basera donc sur ces seuils pour renvoyer le taux de remise à appliquer. Si le total de la facture est différent du montant d'un des seuils (cas général), la fonction renverra le taux de remise correspondant au seuil inférieur le plus proche.

La fonction (construite dans notre exemple en E10) sera donc la suivante :

=RECHERCHEV(E9;remise;2;1)*E9

- **E9** est la cellule contenant le total hors taxes qui est la **valeur_cherchée** dans la *table_matrice*.
- *remise* est le nom de la *table_matrice* (ou plage de cellules) contenant le barème des remises où la recherche sera effectuée.
- 2 est le numéro de la colonne de la *table_matrice* "*remise*" qui contient le taux de remise à appliquer.
- 1 signifie qu'en l'absence de la **valeur_cherchée** dans la *table_matrice*, la valeur inférieure la plus proche sera renvoyée.

Le résultat de la recherche est multiplié par le montant hors taxes en E9 (*E9) pour obtenir le montant de la remise.

Par exemple,
pour un total hors taxes de 6325 € le taux de remise applicable sera donc **5%**

La fonction =CHOISIR()

Exemple avec des taux de remise variables en fonction d'un code (1,2 ou 3) affichant des remises de 5%, 10% ou 15 % :

- La cellule **A1** contient le code d'escompte (1, 2 ou 3),
- La cellule en B1 s'écrit :
`=CHOISIR(A1;5%;10%;15%)`

où Si le code de d'escompte est "1" la valeur de B1 sera 5%, si le code de d'escompte est "2" la valeur de B1 sera 10% et si le code d'escompte est "3", la valeur de B1 sera 15%.

La formule peut contenir autant de variables que vous le désirez.

www.admexcel.com

Les fonctions statistiques

La fonction =NB.SI()

Cette formule permet de compter le nombre de cellules non vides correspondant au critère désiré.

Voici sa syntaxe :

=NB.SI(**Plage**;Critère)

Plage est la plage de cellules sur laquelle on va chercher le nombre de cellules correspondant au critère désiré,
Critère désigne le choix retenu, nombre ou texte.

Exemple : Dans le tableau ci-dessous on désire compter le nombre de réponses "oui".

	A	B
1	Jacques	oui
2	Pierre	non
3	Paul	oui
4	Philippe	oui
5	Stéphane	non
6	Julien	non
7	Maxime	oui
8		

En A10 on écrit la formule suivante :

=NB.SI(B2:B8;"oui")

Le résultat donne : 4

La fonction =NB.SI.ENS()

Compter le nombre de cellules non vides correspondant à plusieurs critères.

Voici sa syntaxe :

=NB.SI.ENS(**Plage_critères1**;Critères1;**[plage_critères2**;critère2];...)

1^{er} exemple :

	A	B	C	D	E	F
1	Nom	Classe	sexe			
2	MATA	CP	Garçon			
3	LOIPT	CE1	Garçon			Garçon
4	TROUNI	CE2	Fille		CP	3
5	NUIERA	CP	Garçon			
6	RATIN	CM1	Fille			
7	GARTIM	CM1	Fille			
8	BAARDI	CP	Garçon			
9	AGRAO	CE2	Fille			
10	ROIT	CP	Fille			
11	LIART	CM2	Garçon			
12	MASE	CM2	Fille			
13	GAREZ	CE1	Garçon			
14						

=NB.SI.ENS(C:C;F3;B:B;E4)

Ici :

- Première plage de critères ; la colonne du sexe, la colonne C (dans la formule **C:C**),
- Premier critère, le sexe saisi en cellule **F3**,
- Deuxième plage de critères, la colonne des classes, la colonne B (dans la formule **B:B**),
- deuxième critère, la classe saisi en cellule **E4**,

2eme exemple :

Compter le nombre de cellules entre 2 montants. Les critères sont obtenus en chaînant dans la formule les symboles « >= » et « <= » avec les références aux cellules servant de critères. Le chaînage se fait avec le symbole &.

	A	B	C	D	E	F	G
1	AUSTIN	Homme	COM	1 800			
2	BOURD	Femme	TEC	1 300		Salaire mini	1800
3	CELAIRE	Femme	ADM	1 450		Salaire maxi	2000
4	CHINOIS	Homme	COM	1 620		Nb de personnes	6
5	DELON	Femme	COM	1 930			
6	DEPARD	Femme	FIN	2 720			
7	DIAL	Homme	ADM	1 845			
8	NEMAR	Homme	ADM	1 730			
9	NOUGA	Homme	TEC	1 640			
10	POURUN	Homme	COM	1 550			
11	TERIEUR	Femme	COM	1 620			
12	TRAINE	Femme	TEC	1 480			
13	TUDEIL	Homme	COM	1 770			
14	ZIER	Homme	TEC	1 690			
15	ZOULOU	Homme	ADM	1 530			
16	ADJANI	Homme	ADM	1 710			
17	CHARME	Femme	FIN	1 980			
18	DAT	Femme	TEC	1 650			
19	DEJEAN	Homme	ADM	1 780			
20	DELUNE	Homme	COM	1 920			
21	DERAND	Homme	ADM	1 810			
22							

Formule en G4 : `=NB.SI.ENS(D:D;">="&G2;D:D;"<="&G3)`

Formule en G4 (dans la boîte de dialogue) : `=NB.SI.ENS(D:D;">="&G2;D:D;"<="&G3)`

Ici :

- Plage de critères ; la colonne des salaires, la colonne D (dans la formule **D:D**),
- Premier critère, le salaire mini saisi en cellule **G2**,
- deuxième critère, le salaire maxi saisi en cellule **G3**,

La formule =NBVAL()

Cette fonction compte le nombre de cellules non vides à l'intérieur d'une plage de cellules spécifiée. On peut l'utiliser, par exemple, pour connaître le nombre de personnes contenues dans une liste.

Voici sa syntaxe :

=NBVAL(plage_de_cellules)

plage_de_cellules peut concerner plusieurs cellules contiguës ou dispersées.

Exemple : Dans le tableau suivant on cherche le nombre de noms figurant dans la liste.

	A	
1		
2		
3	Jacques	
4	Pierre	
5	Paul	
6	Philippe	
7	Stéphane	
8	Julien	
9	Maxime	
10		

En A11 on écrit la formule suivante (volontairement on prendra une plage plus grande que celle contenant les noms) :

=NBVAL(A1:A10)

Le résultat donne : **7**

Les fonctions financières

La fonction =AMORLIN()

Cette formule permet de calculer l'amortissement linéaire d'un bien pour une annuité complète. Pour proratiser la première (et la dernière) annuité il faudra effectuer des calculs complémentaires.

Voici sa syntaxe :

=AMORLIN(**Coût**; *Valeur_rés*; *Durée*)

Coût représente le prix d'achat du bien,

Valeur_rés représente la valeur du bien à l'issue de l'amortissement (généralement 0 (zéro) pour un amortissement comptable),

Durée est la durée d'amortissement en nombre d'années.

Exemple : un bien acheté 10000 € dont la valeur résiduelle sera de zéro est amorti en 5 ans. La formule est la suivante :

=AMORLIN(10000;0;5)

Le résultat donne : **2000**


Activation des macros complémentaires "Utilitaire d'analyse".

Pour Excel XP – 2003


Utilisez le menu "Outils/ Macros complémentaires", cochez l'option "Utilitaire d'analyse" et cliquez sur OK.

Pour Excel 2007

Cliquez sur le bouton Office puis sur "Options Excel"
Cliquez sur la catégorie "Complément"
Puis, dans la liste "Gérer" choisissez "Compléments Excel"
Cliquez sur le bouton "Atteindre"



Cochez les options suivantes :



www.admexcel.com